v.8.1

Programación 1 - Práctica 5, parte 2.

1 Clasificando elementos de una lista

La Práctica 5, primera parte contenía una serie de problemas en los cuales había que *filtrar* los elementos de una lista, quedándose con aquellos que cumplían una determinada condición. Vimos en teoría que la función `filter` nos servía para resolver esta clase de problemas. La signatura de dicha función es la siguiente:

```
; filter : (X -> Boolean) List(X) -> List(X)
```

Dado un predicado `p` y una lista `l` con objetos en `X`, queremos devolver una lista con aquellos objetos de `l` para los cuales `p` evalúa a `#true`.

```
(define (filter p l)
          (cond [(empty? l) empty]
                [else (if (p (first l))
                          (cons (first l) (filter p (rest l)))
                          (filter p (rest l)))]))
```

Algunos ejemplos:

```
(filter even? (list 1 2 3 4 5))
 ==
(list 2 4)
```

```
(filter string? (list 3 "Lista" #true "heterogénea"))
 ==
(list "Lista" "heterogénea")
```

Ejercicio 1. Resuelva los ejercicios 12, 13, 15, 16 y 17 de la Práctica 5, primera parte utilizando `filter`.

Ejercicio 2. Diseñe una función `pares` que tome una lista de números `l` y devuelva una lista con los números pares de `l`.

Ejemplo:

```
pares (list 4 6 3 7 5 0)
```

```
= (list 4 6 0)
```

Ejercicio 3. Diseñe una función `cortas` que tome una lista de strings y devuelva una lista con aquellas palabras de longitud menor a 5.

Ejemplo:

```
cortas (list "Lista" "de" "palabras" "sin" "sentido")
= (list "de" "sin")
```

Ejercicio 4. Diseñe una función `cerca` que tome una lista de puntos del plano (representados mediante estructuras `posn`), y devuelva la lista de aquellos puntos que están a distancia menor a `MAX`, donde `MAX` es una constante de su programa.

Ejemplo (considerando `5` para la constante) :

```
cerca (list (make-posn 3 5) (make-posn 1 2) (make-posn 0 1) (make-posn 5 6))
= (list (make-posn 1 2) (make-posn 0 1))
```

Ejercicio 5. Diseñe una función `positivos` que tome una lista de números y se quede sólo con aquellos que son mayores a 0.

```
(positivos (list -5 37 -23 0 12))
= (list 37 12)
```

Ejercicio 6. Diseñe una función `eliminar-0` que tome una lista de números y devuelva la lista luego de eliminar todas las ocurrencias del número 0.

Ejemplos:

```
(eliminar-0 (list 1 0 0 0 7 6 0))
= (list 1 7 6)
```

```
(eliminar-0 (list 1 2 3))
= (list 1 2 3)
```

Ejercicio* 7. ¿Es posible resolver el ejercicio anterior para cualquier número?. Piense en cómo definir una función que tome una lista de números y un número y devuelva la lista luego de eliminar todas las ocurrencias del segundo argumento.

---

2 Aplicando transformaciones a cada elemento de una lista

En la Práctica 5, primera parte se presentan algunos problemas cuya

solución se obtiene aplicando una determinada transformación a cada uno de los elementos de una lista.

Es decir, dada una función `f`, estamos interesados en transformar la lista:

$$[a0, a1, ..., an]$$

en

$$[f(a0), f(a1), ..., f(an)]$$

En clase de teoría vimos que la función `map` se podía utilizar para resolver problemas de este tipo. Podemos escribir su signatura como sigue:

```
; map : (X -> Y) List(X) -> List(Y)
```

Es decir, dada una función que transforma objetos de `X` en objetos de `Y`, y una lista con objetos en `X`, devuelve una lista con objetos en `Y`. Recordemos la definición vista en clase:

```
(define (map f l)
  (cond [(empty? l) empty]
        [else (cons (f (first l)) (map f (rest l)))]))
```

Con esta definición obtenemos, por ejemplo:

```
(map sqr (list 1 2 3 4 5))
 ==
(list 1 4 9 16 25)
```

```
(map string-length (list "Lista" "de" "palabras"))
 ==
(list 5 2 8)
```

Ejercicio 8. Diseñe la función `raices`, que dada una lista de números, devuelve una lista con las raíces cuadradas de sus elementos.

```
(raices (list 9 16 4))
= (list 3 4 2)
```

Ejercicio 9. Diseñe una función `distancias` que tome una lista de puntos del plano y devuelva una lista con la distancia al origen de cada uno.

Ejemplo:

```
(distancias (list (make-posn 3 4) (make-posn 0 4) (make-posn 12 5)))
```

```
= (list 5 4 13)
```

Ejercicio 10. Diseñe una función `anchos` que tome una lista de imágenes y devuelva una lista con el ancho de cada una.

Ejemplo:

```
(anchos (list (circle 30 "solid" "red") (rectangle 10 30 "outline" "blue")))
= (list 60 10)
```

Ejercicio 11. Diseñe la función `signos`, que dada una lista de números, devuelve una lista con el resultado de aplicarle a cada elemento la función `sgn2` definida en la práctica 1.

```
(signos (list 45 32 -23 0 12))
= (list 1 1 -1 0 1)
```

Ejercicio 12. Diseñe una función `cuadrados` que tome una lista de números y devuelva otra lista donde los elementos que aparezcan sean el cuadrado de los elementos de la lista original.

Ejemplo:

```
(cuadrados (list 1 2 3))  = (list 1 4 9)
```

Ejercicio 13. Diseñe una función `longitudes` que tome una lista de cadenas y devuelva una lista de números que corresponda con la longitud de cada cadena de la lista original.

Ejemplo:

```
(longitudes(list "hola" "cómo" "estás?")) = (list 4 4 6)
```

Ejercicio 14. Diseñe la función `convertirFC`, que convierte una lista de temperaturas medidas en Fahrenheit a una lista de temperaturas medidas en Celsius.

---

3 Operando con los elementos de una lista

Finalmente, algunos ejercicios de la Práctica 5, primera parte buscaban obtener un valor como resultado de realizar una opecion que involucra a todos los elementos de la lista. Es decir, dada una función `f` y una lista

$$[a0, a1, ..., an]$$

estos ejercicios se resolvian haciendo

```
(f a0 (f a1 (... (f an-1 an))))
```

En los ejercicios 25 y 26, dicha función es el producto (`*`) y la concatenación de Strings (`string-append`) respectivamente.

Tal como vimos en teoría, podemos encontrar un patrón en común en la solución de estos ejercicios, y a este patron de solución le llamamos `fold`. La función `fold` recibe tres argumentos:

- La función `f` con la que se quiere operar los elementos de la lista;
- Un valor `c`, que es el resultado esperado para la lista vacía;
- La lista `l` a transformar.

Al evaluarse la expresión

```
(fold f c (cons a0 (cons a1 (... (cons an '())))))
```

se obtiene

```
(f a0 (f a1 (... (f an c))))
```

Detengase y piense en la signatura de la función `fold`.

Algunos ejemplos concretos:

```
(fold * 1 (list 1 2 3 4 5))
 ==
120
```

```
(fold string-append "" (list "Pro" "gra" "ma" "ción."))
 ==
"Programación"
```

Usando *racket*, la definición de `fold` quedaría:

```
(define (fold f c l)
  (cond [(empty? l) c]
        [else (f (first l) (fold f c (rest l)))]))
```

Ejercicio 15. Diseñe una función `prod` que multiplica los elementos de una lista de números. Para la lista vacía, devuelve `1`.

```
(prod (list 1 2 3 4 5))
= 120
```

Ejercicio 16. Diseñe una función `pegar` que dada una lista de strings, devuelve el string que se obtiene de concatenar todos los elementos de la lista.

```
(pegar (list "Las " "lis" "tas " "son " "complicadas" "."))
= "Las listas son complicadas."
```

Ejercicio 17. Diseñe una función `max` que devuelve en máximo de una lista de naturales. Para la lista vacía, devuelve `0`.

```
(max (list 23 543 325 0 75))
= 543
```

---

4 Más ejercicios

Los problemas de esta sección se pueden resolver utilizando las funciones presentadas en las secciones precedentes (en varios de ellos necesitará usar más de una). Intente utlizar `map`, `fold` y `filter` para construir sus soluciones.

Veamos un ejemplo:

Ejercicio 18. Diseñe una función `sumcuad` que dada una lista de números, devuelve la suma de sus cuadrados. Para la lista vacía, devuelve `0`.

Dada una lista `l`, podemos dividir este problema en dos tareas:

- Calcular los cuadrados de todos los elementos de `l`, y
- sumar estos valores.

Diseñamos una solución para cada tarea:

```
; cuadrados : ListN -> ListN
; calcula los cuadrados de todos los elementos de una lista de números
(check-expect (cuadrados (list 1 2 3 4 5)) (list 1 4 9 16 25))
(check-expect (cuadrados empty) empty)
(check-expect (cuadrados (list 11 13 9)) (list 121 169 81))
(define (cuadrados l) (map sqr l))
```

```
; suma : ListN -> Number
; suma todos los elementos de una lista de números
(check-expect (suma (list 1 2 3 4 5)) 15)
(check-expect (suma empty) 0)
(check-expect (suma (list 11 13 9)) 33)
(define (suma l) (fold + 0 l))
```

Ahora podemos simplemente combinar ambas partes para resolver el problema:

```
; sumcuad : ListN -> Number
; suma los cuadrados de una lista de números
(check-expect (sumcuad (list 1 2 3 4 5)) 55)
(check-expect (sumcuad empty) 0)
(check-expect (sumcuad (list 11 13 9)) 371)
(define (sumcuad l) (suma (cuadrados l)))
```

La idea es entonces que la función a definir se pueda construir combinando otras más sencillas sobre listas, y que cada una de estas últimas se puedan definir usando `map`, `fold` y `filter`.

> Para `map`, `fold` y `filter` puedes usar las definiciones que vimos en esta práctica o utilizar las funciones que vienen con *DrRacket* . Si elige esta última opción, debes tener en cuenta dos cosas: 1) Debes cargar el lenguaje *Estudiante Intermedio* 2) En *DrRacket* la función incluida para `fold` se llama **`foldr`**.

Ejercicio 19. Diseñe la función `sumdist`, que dada una lista `l` de estructuras `posn`, devuelve la suma de las distancias al origen de cada elemento de `l`.

Ejemplo:

```
(sumdist (list (make-posn 3 4) (make-posn 0 3)))
= 8
```

Ejercicio 20. Diseñe una función `multPos`, que dada una lista de números `l`, multiplique entre sí los números positivos de `l`.

Ejemplo:

```
(multPos (list 3 -2 4 0 1 -5))
= 12
```

Ejercicio 21. Diseñe una función `sumAbs`, que dada una lista de números, devuelve la suma de sus valores absolutos.

Ejemplo:

```
(sumAbs (list 3 -2 4 0 1 -5))
= 15
```

Ejercicio 22. Diseñe la función `raices`, que dada una lista de números `l`, devuelve una lista con las raíces cuadradas de los números no negativos de `l`.

Ejemplo:

```
(raices (list 16 -4 9 0))
= (list 4 3 0)
```

Ejercicio 23. Diseñe la función `saa`, que dada una lista de imágenes, devuelva la suma de las áreas de aquellas imágenes `"Anchas"`.

Ejemplo:

```
(saa (list (circle 20 "solid" "red")
           (rectangle 40 20 "solid" "blue")
           (rectangle 10 20 "solid" "yellow")
           (rectangle 30 20 "solid" "green")))
= 1400
```

Ejercicio 24. Diseñe la función `algun-pos`, que toma una lista de listas de números y devuelve `#true` si y sólo si para alguna lista la suma de sus elementos es positiva.

Ejemplos:

```
(algun-pos (list (list 1 3 -4 -2) (list 1 2 3 -5) (list 4 -9 -7 8 -3)))
= #true
(algun-pos (list empty (list 1 2 3)))
= #true
(algun-pos (list (list -1 2 -3 4 -5) empty (list -3 -4)))
= #false
```

Ejercicio 25. Diseñe la función `long-lists`, que toma una lista de listas y devuelve `#true` si y sólo si las longitudes de todas las sublistas son mayores a 4.

Ejemplos:

```
(long-lists (list (list 1 2 3 4 5) (list 1 2 3 4 5 6) (list 87 73 78 83 33)))
= #true
(long-lists (list '() '() (list 1 2 3)))
= #false
(long-lists (list (list 1 2 3 4 5) empty))
= #false
```

Ejercicio 26. Diseñe una función `todos-true` que toma una lista de valores de cualquier tipo, y devuelve `#true` si y sólo si todos los valores booleanos de la lista son verdaderos. Caso contrario, devuelve `#false`.

Ejemplos:

```
(todos-true (list 5 #true "abc" #true "def"))
= #true
```

```
(todos-true (list 1 #true (circle 10 "solid" "red") -12 #false))
= #false
```

Presente al menos dos ejemplos más en su diseño.

Ejercicio 27. Dada la definición de la estructura `alumno`:

```
(define-struct alumno [nombre nota faltas])
; alumno  (String, Number, Natural). Interpretación
; - nombre representa el nombre del alumno.
; - nota representa la calificación obtenida por el alumno (entre 0 y 10).
; - faltas: número de clases a las el alumno no asistió.
```

Diseñe las siguientes funciones:

- `destacados`, que dada una lista de alumnos, devuelve una lista con el nombre de aquellos alumnos que sacaron una nota mayor o igual a 9.

  Ejemplo:

  ```
  (destacados (list (make-alumno "Ada Lovelace" 10 20)
                    (make-alumno "Carlos Software" 3.5 12)))
   = (list "Ada Lovelace")
  ```

- `condicion`, que dado un alumno, determine su condición de acuerdo a las siguientes reglas:

  - si la nota es mayor o igual a 8, su condición es `"promovido"`.
  - Si la nota es menor a 6, su condición es `"libre"`.
  - En cualquier otro caso, la condición es `"regular"`.

- `exito`, que dada una lista de alumnos, devuelve `#true` si ninguno está libre. Caso contrario, devuelve `#false`.

  Ejemplo:

  ```
  (exito (list (make-alumno "Juan Computación" 5 13)
               (make-alumno "Carlos Software" 3.5 12)
               (make-alumno "Ada Lovelace" 10 20)))
  = #false
  ```

- `faltas-regulares`, que dada una lista de alumnos, devuelve la suma de las ausencias de los alumnos regulares.

  Ejemplo:

```
(faltas-regulares (list (make-alumno "Juan Computación" 7 2)
                        (make-cliente "Carlos Software" 3.5 4)
                        (make-alumno "Ada Lovelace" 10 1)))

= 2
```

- `promovidos-ausentes`, que dada una lista de alumnos, devuelve una lista con el nombre de aquellos alumnos promovidos que no asistieron a tres o más clases.

  Ejemplo:

```
(promovidos-ausentes (list (make-alumno "Juan Computación" 9 3)
                        (make-cliente "Carlos Software" 3.5 2)
                        (make-alumno "Ada Lovelace" 10 1)))
= (list "Juan Computación")
```